# $Id: PKGBUILD 144377 2011-12-05 12:49:24Z jgc $
# Maintainer: Jan de Groot <jgc@archlinux.org>
# Contributor: Thomas Baechler <thomas@archlinux.org>
# Contributor: Tom Newsom <Jeepster@gmx.co.uk>

pkgname=libdvdread
pkgver=4.2.0
pkgrel=1
pkgdesc="Provides a simple foundation for reading DVD video disks"
arch=(i686 x86_64)
url="http://dvdnav.mplayerhq.hu"
license=('GPL')
depends=('glibc')
makedepends=('libdvdcss')
optdepends=('libdvdcss: for decoding encrypted DVDs')
options=('!libtool')
source=(http://dvdnav.mplayerhq.hu/releases/${pkgname}-${pkgver}.tar.bz2)
sha256sums=('0bea15da842a4b04a482b009d72dcc6d9c9524ccc1bf67e5748319ec5ada8097')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  # http://lists.mplayerhq.hu/pipermail/dvdnav-discuss/2009-February/000986.html (FS#19324)

  ./autogen.sh --prefix=/usr
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
